

dREL Language Specifications

Appendix to the paper:

dREL: A RELATIONAL EXPRESSION LANGUAGE FOR DICTIONARY METHODS

Nick Spadaccini, Ian R. Castleden, Doug du Boulay, Sydney R. Hall

School of Biomedical, Biomolecular and Chemical Sciences, The University of Western Australia, Nedlands 6009, Australia

Email: Nick.Spadaccini@uwa.edu.au, Sydney.Hall@uwa.edu.au

Primitive Data Types

dREL supports the following primitive data types of the *values* for variables appearing in methods expressions. Local variable names (as opposed to global data tags) are restricted to alphanumeric characters only.

- Character strings
- Integer numbers
- Real numbers
- Complex numbers
- Measured numbers

Data typing may be achieved by explicitly within the dictionary definitions of the object, or implicitly from usage in an expression, or explicitly using a function. DDLm dictionary definitions specify data types using the `TYPE` attributes (see `_type.contents`, `_type.container`, `_type.purpose`, `_type.dimension`).

1. CHARACTER STRINGS

1.1 Dictionary definition

The data dictionary specifies the type of a *data tag* using the `TYPE` attribute `_type.contents`.

1.2 Inline definition

Character strings are created by enclosing a string in quoting literals. Matching single and double quote characters at the extremities of a single line string implicitly identify a literal object as `TYPE CHARACTER`. Matching triple quote characters at the extremities of a multi-line string implicitly identify a literal object as `TYPE CHARACTER`.

1.2.1 Single quotes

Matching single quote characters at the extremities of a *single line* string implicitly identify a literal object as `TYPE character`. The following is simple character string.

```
'single quotes make it easy to embed a "double quote"'
```

1.2.2 Double quotes

Matching double quote characters at the extremities of a *single line* string implicitly identify a literal object as `TYPE character`. The following is simple character string.

```
"double quotes make it easy to embed a 'single quote'"
```

It is also possible to use *C-style elides* to achieve this effect.

```
"double quotes don't prevent the use of a \"double quote\""
```

1.2.3 Triple quotes

Matching triple quote characters at the extremities of a *multi-line* string implicitly identify a literal object as `TYPE` character. The following is simple character string.

```
""" triple quotes  
are  
multi-line"""
```

This is equivalent to

```
"triple quotes\nare\nmulti line\n"
```

Triple quotes comprised of the single quote literal are also supported.

```
'''single or double quotes are can be  
used to define the triple quote sequence.'''
```

2. INTEGER NUMBERS

dREL supports decimal and hexadecimal *Integer* numbers. These are identified in three ways; explicitly from dictionary definitions of the object, implicitly from usage in the expression language, or explicitly using a function.

2.1 Dictionary definition

The data dictionary specifies the type of a *data tag* using the `TYPE` attribute `_type.contents`.

2.2 Inline definition

2.2.1 Decimal integers

The syntax of a decimal integer is: $[+-]?[0-9]^+$
An example decimal integer is: -23

2.2.2 Hexadecimal integers

The syntax of a hexadecimal integer is: $[0][xX][0-9a-fA-F]^+$
An example hexadecimal integer is: 0x6672af

3. REAL NUMBERS

dREL supports decimal and scientific *Real* (or floating-point) objects. *Real* numbers are identified in three ways; explicitly from dictionary definitions of the object, implicitly from usage in the expression language, or explicitly using a function.

3.1 Dictionary definition

The data dictionary specifies the type of a *data tag* using the `TYPE` attribute `_type.contents`.

3.2 Inline definition

3.2.1 Decimal real numbers

The syntax of a decimal real number is: $[+-]?([0-9]^+.[0-9]^+|.[0-9]^+)[[Ee][+-]?[0-9]^+]?$
An example decimal real number is: -7893.8221 or -7.89382e+3

3.3 Explicit definition

Conversion to real number is achieved with the function:

- `Float()`

4. COMPLEX NUMBERS

dREL supports *Complex* number objects. *Complex* numbers are identified in three ways; explicitly from dictionary definitions of the object, implicitly from usage in the expression language, or explicitly using a function.

4.1 Dictionary definition

The data dictionary specifies the type of a *data tag* using the TYPE attribute `_type.contents`.

4.2 Explicit definition

Conversion to a complex number is achieved with the function:

- *Complex* (*Nreal*, *Nimag*)

5. MEASURED NUMBERS

A *Measured value* consists of a number and its standard uncertainty appended in parentheses. The uncertainty value is an integer scaled to the precision of the last digits of the measurement value. *Measurement* numbers are identified in three ways; explicitly from dictionary definitions of the object, implicitly from usage in the expression language, or explicitly using a function.

5.1 Dictionary definition

The dictionary definitions declare the TYPE of a data tag with the following set of attribute declarations:

<code>_type.contents</code>	Real
<code>_type.purpose</code>	Measured

The value of the attribute `_type.contents` can also be *Integer* or *Complex*.

5.2 Inline definition

5.2.1 Measured numbers

The syntax of a measurement number is: `[Real|DecimalInteger]\([[-9][0-9]*\)`

An example measurement number is: `-783.2(12) = -783.2±1.2`

Other examples are `x.xxE-yy(zz)` or `x.xx(zz)E-yy` or `x.xxE-yy(z.zzE+ww)` where a ‘.’ in the standard uncertainty value indicates an explicit value.

5.3 Explicit definition

Conversion to a measurement number is achieved with the function:

- *Measure* (*val*, *su*)

Container Types for dREL

dREL supports the container types

- List *List data is bounded by square brackets []*
- Array *Array data is bounded by square brackets []*
- Matrix *Matrix data is bounded by square brackets []*
- Table *Table data is bounded by curly brackets { }*
- Single

dREL also supports the nesting and mixing of container types i.e. the definition

<code>_type.container</code>	List
------------------------------	------

```

_type.contents          Real
-type.dimension         (5 (3))

```

refers to a list of five arrays; each array contains three real numbers.

1. LIST CONTAINERS

List containers are objects with the following properties.

- **Type:** contained items may be of any, but the same, TYPE.
- **Dimension:** Lists are single dimensioned.
- **Size:** the length of a list need *not* pre-defined.
- **Access:** indexed by integers (*implied starting index is 0*).
- **Shape:** bounded by [...] and may be nested.

Lists are created in three ways; explicitly from dictionary definitions of the object, implicitly from usage in the expression language, or explicitly using a function.

1.1 Dictionary definition

The dictionary definitions declare the nature of a *List* container with attribute declarations. Here are such declarations for a list of real numbers of nine elements.

```

_type.container        List
_type.contents         Integer

```

1.2 Inline definition

Lists may be defined inline using the *List(...)* function. E.g.

```
List([ 1, 7, 3, 10 ]) which is also implied by [1,7,3,10]
```

2. ARRAY CONTAINERS

Array containers are objects with the following properties.

- **Type:** may contain items of *any* TYPE.
- **Dimension:** are single or multi dimensioned.
- **Size:** predetermined upper extents; minimum elements assumed to be 1.
- **Access:** indexed by integers (*implied starting index is 0*).
- **Shape:** bounded by [...] and may be nested.

Arrays are created in three ways; explicitly from dictionary definitions of the object, implicitly from usage in the expression language, or explicitly using a function.

2.1 Dictionary definition

The dictionary definitions declare the nature of a *Array* container with attribute declarations. Here are the attributes for defining an array of binary numbers.

```

_type.container        Array
_type.contents         Binary

```

2.2 Inline definition

Vectors may be defined inline using the *Array(...)* function. E.g.

```
Array([[126,255,0],[123,245,10]]) which is also implied by [[126,255,0],[123,245,10]]
for a data item defined as _type.container Array
```

3. MATRIX CONTAINERS

Matrix containers are objects with the following properties.

- **Type:** only contain items of *number* TYPE.
- **Dimension:** are single or multi dimensioned.
- **Size:** predetermined upper extents; minimum elements assumed to be 1.
- **Access:** indexed by integers (*implied starting index is 0*).
- **Shape:** bounded by [...] and may be nested.

Matrices are created in three ways; explicitly from dictionary definitions of the object, implicitly from usage in the expression language, or explicitly using a function.

3.1 Dictionary definition

The dictionary definitions declare the nature of a *Matrix* container with attribute declarations. Here are the attributes for defining a vector of three real numbers, indexed from 0 to 2.

```

_type.container      Matrix
_type.contents      Real
_type.dimension     [3]

```

3.2 Inline definition

Vectors may be defined inline using the *Matrix(...)* function. E.g.

`Matrix([10.2, 12.3, 7.4])` which is also implied by `[10.2,12.3,7.4]` for a data item defined as `_type.container Matrix`

4. TABLE CONTAINERS

Table containers are similar to *Lists* except that each *value* in the table may have an *associated key*. A table has the following properties.

- **Type:** can contain values of any, but the same, TYPE.
- **Dimension:** single dimensioned list; each "*key*":*val* is considered as one element.
- **Size:** the length of a table is *not* pre-determined.
- **Access:** by key; the default keys are sequential integers starting at 0.
- **Shape:** bounded by {...} and may be nested.

Tables are created in two ways; explicitly from dictionary definitions of the object, implicitly from usage in the expression language, or explicitly using a function.

4.1 Dictionary definition

The dictionary definitions identify a *Table* object with the following attribute declarations.

```

_type.container      Table
_type.contents      Real

```

A *Table* differs from a *List* in several important ways. A *List* object contains a specified number of values that are identified explicitly by sequence. A *Table* contains a sequence of character or number values which identified by a key.

4.2 Explicit definition

Conversion of a sequence of objects to a new list is achieved with the function *Table ('key':val,..)*. E.g.

`Table("left":"links","right":"recht")` implied by `{"left":"links","right":"recht"}`

5. SINGLE CONTAINERS

Single containers are a single value with the following properties.

- **Type:** may be of any TYPE.
- **Dimension:** a single value.

- **Size:** 1.

Single values are created in three ways; explicitly from dictionary definitions of the object, implicitly from usage in the expression language, or explicitly using a function.

5.1 Dictionary definition

The dictionary definitions declare the nature of a *Single* container with attribute declarations. Here is a declaration for a real number.

```
_type.container      Single
_type.contents      Real
```

5.2 Inline definition

Single values may be specified inline by equating it to another single value. E.g.

```
a = 5.
z = a
```

Language Basics

In this section the basic syntax of dREL, and the language elements that lead up to controlling the execution flow, are introduced. It is important to appreciate that dREL does not support, or require, data declarations other than those already discussed. Nor does it provide input/output control statements.

1. ASSIGNMENT EXPRESSIONS

1.1 Named objects

A NAMED object or “variable” in dREL may only be created on assignment. The *typing* of a variable is by coercion. The *scope* of a variable is *local*.

1.2 Assignment statements

1.2.1 The process of object transfer is initiated with the “=” character which transfers the value of the right-hand expression of objects *Robjcts* to the left-hand objects *Lobjcts*. The general form of the object transfer is:

$$Lobjcts = Robjcts \quad \text{or} \quad Lobjcts = \{ \text{multi-line expression} \}$$

In the example below the value of the *literal* Integer object, “5”, is assigned to a mutable NAME object, the variable string “x”.

```
x = 5
```

Robjcts may also be an expression of objects.

```
x = y * z
y = Sin (a) + Cos (a)
```

Multiple transfers are also allowed.

```
a, b, c = 3.628, -7.67, 5.329
```

1.2.2 The process of object incrementation is initiated with the “+=” digraph which increments the values of the right-hand expression of objects *Robjcts* to the left-hand objects *Lobjcts*. The general form of an object incrementation is:

$$Lobjcts += Robjcts$$

In the example below the value of the *literal* Integer “1”, is added to the existing value in a mutable NAME object, the single variable “x”.

`x += 2` *i.e. if the value of x is initially 5, becomes 7*

Lobjects may also be a multi-element container whereas *Robjects* may be either a single value or a multi-element container E.g.

`vect += 1` *i.e. if vect is initially [3,3,3], becomes [4,4,4]*
`vect += [1,2,3]` *i.e. if vect is initially [3,3,3], becomes [4,5,6]*

1.2.2 The process of object decrementation is initiated with the “-=” digraph which decrements the values of the right-hand expression of objects *Robjects* to the left-hand objects *Lobjects*. The general form of an object incrementation is:

Lobjects -= Robjects

In the example below the value of the *literal* Integer “1”, is subtracted from the existing value in a mutable NAME object, the single variable “x”.

`x -= 2` *i.e. if the value of x is initially 5, becomes 3*

Lobjects may also be a multi-element container whereas *Robjects* may be either a single value or a multi-element container E.g.

`vect -= 1` *i.e. if vect is initially [3,3,3], becomes [2,2,2]*
`vect -= [1,2,3]` *i.e. if vect is initially [3,3,3], becomes [2,1,0]*

1.2.3 The process of object appending is initiated with the “+=” trigraph which appends the values of the right-hand expression of objects *Robjects* to the end of left-hand objects *Lobjects*. The general form of an object appending is:

Lobjects += Robjects

Lobjects must be a multi-element container whereas *Robjects* may be either a single value or a multi-element container E.g.

`vect += 1` *i.e. if vect is initially [3,3,3], becomes [3,3,3,1]*
`vec2 += [1,2,3]` *i.e. if vec2 is initially [3,3,3], becomes [3,3,3,1,2,3]*
`matx += [[1,2,3]]` *i.e. if matx is initially [[3,2,1]], becomes [[3,2,1],[1,2,3]]*

1.3 Assignment TYPING

In dREL object types are not declared. We have already seen that the typing of *Robjects* items may be determined from dictionary definitions, inline typing constructions or simply inferred by association with objects of known type. The TYPE of *Lobjects* may be set by the same mechanisms, or result directly from the inferred type of the *Robjects* value.

It follows that the statement

`x = 5`

sets the TYPE of “x” as Integer. A new assignment of “x” in the next statement

`x = 10`

is permitted because it has a consistent TYPE. However, the assignment

`x = "Hello World"`

is illegal but will not cause an error message to be raised.

This is contrary to the practice of some scripting languages, but it avoids the faulty and misleading construction of expressions.

2. TYPE COERCION RULES

Type coercion rules are needed when *Objects* expressions contain objects of mixed type. dREL uses the following coercion rule, in order of increasing priority.

Integer → Real → Complex

3. COMMENTS

Comments are non-executable strings. In dREL a sequence of characters following an unquoted *sharp* or *hash* symbol # is interpreted as a comment, up to the end-of-line character. Here are typical examples.

```
x = 5      # a comment follows an in-line hash
```

The following statement does not contain a comment because the hash symbol is contained within a quoted string.

```
s = "# this is *not* a comment"
```

4. EXPRESSION OPERATORS AND TERMINATORS

dREL supports the following **arithmetic expression operators**

+	<i>addition</i>
*	<i>product (dot product when applied to vectors)</i>
^	<i>cross product of vectors</i>
**	<i>power of</i>
-	<i>subtraction</i>
/	<i>division</i>

The operands apply to *Integer*, *Real* and *Complex* number objects. They are also applicable to the containers *List* and *Table* provided the elements of these are of TYPE *number*. The expression operators + and * have meaning for manipulating character strings.

dREL supports the following **logical expression operators**

==	<i>equals</i>
!=	<i>not equals</i>
>	<i>greater than</i>
<	<i>less than</i>
>=	<i>greater than or equals</i>
<=	<i>less than or equals</i>
and	<i>and</i>
or	<i>or</i>
not	<i>not</i>
in	<i>matches element of the list</i>
not in	<i>does not matches element of the list</i>

dREL supports the following **expression terminators**

;	<i>semicolon separates multiple expressions in a line</i>
\n	<i>newline closes a line unless a balancing ')', '}' or ']' is missing</i>

Example statements using these terminators follow.

```
a = 234 ; y = 45 ; z = -2
b = (y + z)/2.0
```


$$c = (45 + 72 * (93 + 4) + z)$$

5. SUPPORTED ESCAPE SEQUENCES

The following special character sequences are supported in dREL expressions. Note that the same diagraphs may be used for other purposes in data values, but within the literal dREL scripts the following meanings will be assumed.

<code>\n</code>	<i>newline</i>
<code>\r</code>	<i>carriage return</i>
<code>\f</code>	<i>formfeed</i>
<code>\t</code>	<i>horizontal tab</i>
<code>\x</code>	<i>hexadecimal bit pattern</i>
<code>\0</code>	<i>null character</i>
<code>\\</code>	<i>backslash (\)</i>

Flow Control

dREL supports a range of standard and specialised flow control statements and terminators for controlling the repeated execution of object expressions. These are as follows:

- Do
- Repeat
- For
- Loop
- With
- Break
- Next
- If/Elseif/Else

The essential constituents of a repetitive execution sequence, is as follows.

```
repeat-statement    {
                    *expression block*
                    repeat-terminator (optional)
                    }
```

If more than one expression exists within the expression block, it **MUST** be enclosed within a set of braces "{" and "}". If only one expression is repeated, its association with the *repeat_statement* is implied and the braces are optional. In general, it is good and safe programming practice to always use braces to bound the repeated expression block.

1. DO STATEMENT

Indexed repetition of expressions is supplied with a *Do* statement.

```
Do index = first, last, incr    { *expression block* }
```

The *index* variable is initialised with the *first* index value (or variable) and executes the expression block provided index is less than or equal to the *last* index value (or variable). The *index* is incremented by the *incr* value **AFTER** each execution of the expression block. The *incr* value is option and has a default value of 1.

A typical application of the Do operator follows.

```
Do i = 0,20,2 { total = total + subtotal[i]; }
```

2. REPEAT STATEMENT

Unindexed repetition of expressions is supplied with a *Repeat* statement.

```
Repeat { *expression block* }
```

The expression block MUST contain one or more invocations of the Break statement in order to exit the repeat loop. Repeat loops may be nested. A typical application of the Repeat operator follows.

```
Repeat { i=i+1; if(i>100) Break;... }
```

3. FOR STATEMENT

Manipulation of List items is provided with with a *For* statement.

```
For a in list : n op m { * expression block * }
```

where *a* is the current element of the entire *list*. An optional expression “:*n op m*” is available to control the accessing of list packets, where *n* is the index (starting at 0) for each packet; *op* is the test operator (< > <= >= allowed) and *m* is the test integer operand. The *op* and *m* entries are optional. The index *n* is a local variable and may be tested elsewhere in the script.

An example where *list* is a literal object follows.

```
i = 0
For a in ["Mon", "Tues", "Wednes", "Thurs", "Fri"] {
    Day[i] = a + "day"; i += 1; }
```

4. LOOP STATEMENT

A fundamental function of dREL is to apply and derive data in a *data file* using definitions in a dictionary. Much of this data is in looped lists, and, consequently, there needs to be a simple and transparent way to identify and apply repetitive data items. Data items in the same list are, according to the dictionary language DDLm, classified as belonging to the same generic category group. The id code of a category is therefore a convenient tag to identify groups of items, and to access “packets” (i.e. sub-lists) of data items in lists. The Loop repetition operator is provided primarily for this purpose.

```
Loop local as list : n op m { * expression block * }
```

The string *local* is an object variable, local only to the specific methods script in which it is invoked, which assumes the successive values of *list* during the repeated execution of an expression block. If *list* is a category id code, then the *local* object contains successive sub-list of tagged values (i.e. an implicit *Table*) and individual data items may be accessed as *object attributes* of *local*. An optional expression “:*n op m*” is available to control the looping of list packets, where *n* is the loop index (starting at 0) for each packet; *op* is the test operator (< > <= >= allowed) and *m* is the test integer operand. The *op* and *m* entries are optional. The index *n* is a local variable and may be tested elsewhere in the script.

4.1 Data Loop Example 1

A simple invocation of *Loop* will now be considered for data. This example will access two data items in the category POSITION, known by their data names as `position.vector_xyz` and `position.object_id`. An abbreviated definition of the category and these items follow. Note that `position.object_id` is specified as the category key to each packet of these items.

<code>_category.id</code>	POSITION
<code>category.keyid</code>	' <code>_position.object_id</code> '
<code>_definition.id</code>	'<code>_position.number</code>'
<code>_name.category_id</code>	position
<code>_name.object_id</code>	number
<code>_type.container</code>	Single
<code>_type.contents</code>	Integer
<code>_type.purpose</code>	Index
<code>_definition.id</code>	'<code>_position.object_id</code>'
<code>_name.category_id</code>	position
<code>_name.object_id</code>	object_id
<code>_type.container</code>	Single
<code>_type.contents</code>	Uchar
<code>_definition.id</code>	'<code>_position.vector_xyz</code>'
<code>_name.category_id</code>	position
<code>_name.object_id</code>	vector_xyz
<code>_type.container</code>	Matrix
<code>_type.contents</code>	Real
<code>_type.dimension</code>	[3]

In a data file these items might appear in a looped list (abbreviated) as follows.

```

loop_
  _position.number
  _position.object_id
  _position.vector_xyz
    1      origin   [0.0, 0.0, 0.0]
    2      body-diagonal [5.0, 5.0, 5.0]
    32     diagonal-terminal [10.0, 10.0, 10.0]

```

In a dREL script the *Loop* construct allows individual items in a packet (in this instance the packet contains three values) to be addressed by the *extension name* defined in the dictionary with the attribute `_item.extension` (i.e. `number`, `object_id` and `vector_xyz`).

```

Loop a as position {
  If (a.object_id == "origin") {
    CoordOrigin      = a.vector_xyz  }
  Else
    LocalPosn[a.number] = a.vector_xyz
}

```

4.2 Data Loop Example 2

Another example is needed to illustrate the functionality of the *Loop* operator when handling lists of data from non-hierarchically-related but derived, categories. The prototype dictionary language allows hierarchical relationships between data items to be defined, via category definitions, and these provide access "pathways" which are independent of how these related data are stored in the data file. For instance, items in the same category, or in hierarchically related categories, may be accessed as an attribute extension of either the name of the "parent" category (i.e. the highest category in the family hierarchy) or the name of the hierarchically related category.

All data in a looped list be of the *same category family*. Items from hierarchically related categories may be in more than one looped list but for the purposes of access, the dREL parser subsumes these items into a common list.

However, categories of data that are derived from another category will often use category keys which refer to the same quantities. In these cases, the keys are not implicitly equivalent (as would be

the case if the categories were hierarchically related) but they are “linked” using the DDL attribute `name.parent_item_id`. Here is the definition of an item in the category `GEOM` which is linked to a category key in the category `POSITION` (see Example 1).

```

_definition_id      '_geom.vertex1_id'
_name.category_id   geom
_name.object_id     vertex1_id
_name.linked_item_id '_position.object_id'
_type.container     Single
_type.contents      Text

```

The `name.linked_item_id` attribute specifies that `geom.vertex1_id` has a value that is common to one of the unique values of the item `position.object_id`. This linkage implies that `position.object_id` is a “key” unique item in the category `POSITION`. The same relationships also apply for the items `geom.vertex2_id` and `geom.vertex3_id`, which are shown below in an example data list.

```

loop_
  _geom.type
  _geom.vertex1_id
  _geom.vertex2_id
  _geom.vertex3_id
  point      origin      .      .
  line       origin      body-diagonal  .
  line       body-diagonal diagonal-terminal .
  triangle   origin      body-diagonal  diagonal-terminal

```

Specific values in this list can be accessed via their unique extension names. However, because of the defined relationship between the vertex ID's and the `position.object_id` (in Example 1), these can be used to “point” to specific packets and items in the `POSITION` category using the `<category>[<key>].<extension>` construction. The `With` command used in the example dREL script below is described in the next section and the list-append operator “`++=`” is described below.

```

With p as position
Loop g as geom {
  If (g.type == "point") {
    PointList += Tuple(Tuple(g.vertex1_id,
                           p[g.vertex1_id].vector_xyz))
  }
  Else if (g.type == "line") {
    LineList += Tuple(Tuple(g.vertex1_id, g.vertex2_id),
                     Tuple(p[g.vertex1_id].vector_xyz,
                           p[g.vertex2_id].vector_xyz))
  }
}

```

This illustrates how values from the category list can be directly accessed simply by appending the name extensions to the item which is linked to the key of that list. Executing this script results in the following values strings:

```

PointList[0] is ("origin", [0.,0.,0.])
LineList[0]  is ("origin", "body-diagonal"), ([0.,0.,0.], [5.,5.,5.])
LineList[1]  is ("body-diagonal", "diagonal-terminal"), ([5.,5.,5.], [10.,10.,10.])

```

5. WITH STATEMENT

The `With` statement is identical to the `Loop` statement except that the list pointer is not incremented. This statement is used only to identify the current `list` object within scope and context as a `local` object. The general form is as follows.

```

With local as list { *expression block* }

```

This statement is very useful for accessing data items in the *current packet* of a category lists. This enables items in a list to be addressed as name extension attributes, just as in *Loop*.

```
With p as atom_site
  If (label == p.id) x = p.frac_vector
```

Note the braces about the expression block are required for multiline expressions.

6. **BREAK TERMINATOR**

Repetitive blocks can be exited prematurely with the *Break* keyword. The general form of the statement is as follows.

Break

For example, in the sequence

```
Do i=1:10 {
  Do j=i+1:10 {
    If (a[i] < a[j]) Break
  }
}
```

7. **NEXT TERMINATOR**

Repetitive blocks can be reset prematurely with the *Next* keyword. The general form of the statement is as follows.

Next

For example, in the sequence

```
Do i=1:10 {
  Do j=i+1:10 {
    If (a[i] < a[j]) Next
  }
}
```

8. **IF/ELSEIF/ELSE STATEMENTS**

The standard *If/ElseIf/Else* statements have the following form and sequence. The *If* statement must precede all others in the sequence. The *Else* statement must, if used, follow all others. There may be any number of *ElseIf* statements.

```
If (expr)      { *expression block* }
Else If (expr) { *expression block* }
Else           { *expression block* }
```

Braces around the expression blocks are necessary if they contain more than one statement.

Intrinsic Functions

dREL has an extensive set of intrinsic functions, which are listed in this section according to the following classes.

- CONVERSION and MANIPULATION
- TRIGONOMETRIC
- MATHEMATICAL
- DISCIPLINE

1. CONVERSION AND MANIPULATION FUNCTIONS.

These functions are responsible for fixing the TYPE of the contained object.

<i>Complex()</i>	Convert two arguments (<i>Real</i> , <i>Imag</i>) into a <i>Complex</i> number
<i>Real()</i> , <i>Imag()</i>	Returns <i>real</i> and <i>imaginary</i> part of <i>Complex</i> argument
<i>Integer()</i>	Convert argument into an <i>integer</i> number
<i>Float()</i> , <i>Rem()</i>	Convert to <i>real</i> number, get <i>remainder</i> of real number
<i>Int()</i> , <i>Nint()</i>	Convert to truncated integer, rounded-up integer value
<i>List()</i>	Convert arguments into a <i>List</i> object.
<i>Table()</i>	Convert arguments into a <i>Table</i> object.
<i>Numb()</i>	Convert the character argument into the ascii number equivalent.
<i>Char()</i>	Convert the ascii number argument into a character equivalent.
<i>Minor()</i>	Generate a matrix of <i>minor</i> elements from the matrix argument.
<i>Cofactor()</i>	Generate a matrix of <i>cofactor</i> elements from the matrix argument.
<i>Adjoint()</i>	Generate a matrix of <i>adjoint</i> elements from the matrix argument.
<i>Inverse()</i>	Generate a matrix of <i>inverse</i> elements from the matrix argument.
<i>Transpose()</i>	Generate a matrix of <i>transposed</i> elements from the matrix argument.
<i>Eigen()</i>	Get eigenvalues and vectors of a 3x3 matrix and return as three lists containing four elements (value plus vector of direction cosines).

2. TRIGONOMETRIC FUNCTIONS.

These functions are responsible for performing trigonometric operations on the argument.

<i>Sin()</i> , <i>Cos()</i> , <i>Tan()</i>	Sine, cosine and tangent functions of <i>radian</i> arguments.
<i>Sind()</i> , <i>Cosd()</i> , <i>Tand()</i>	Sine, cosine and tangent functions of <i>degree</i> arguments.
<i>Asin()</i> , <i>Acos()</i> , <i>Atan()</i>	Arcsine, cosine and tangent functions as <i>radians</i> .
<i>Arcsin()</i> , <i>Arccos()</i> , <i>Arctan()</i>	Arcsine, cosine and tangent functions as <i>radians</i> .
<i>Asind()</i> , <i>Acosd()</i> , <i>Atand()</i>	Arcsine, cosine and tangent functions as <i>degrees</i> .
<i>Atan2(a,b)</i> , <i>Atan2d(a,b)</i>	Arctangent function in <i>radians</i> and <i>degrees</i>
<i>Phase()</i>	Get the phase in radians for a <i>Complex</i> number.
<i>Exp()</i> , <i>ExpIm()</i> , <i>ExpImag()</i>	Exponential functions with <i>Real</i> and <i>Complex</i> arguments.
<i>Log()</i> , <i>Ln()</i>	Base-10 and natural logarithm functions.
<i>Pi</i> , <i>TwoPi</i>	Values of π and 2π

3. MATHEMATICAL FUNCTIONS

These functions are responsible for performing mathematical operations on the arguments.

<i>Sqrt()</i>	Get square root of number.
<i>Mod()</i>	Modulus of <i>arg1</i> to base <i>arg2</i> .
<i>Abs()</i> , <i>Magn()</i>	Absolute value of the argument.
<i>Sign()</i>	Sign of argument 2 applied to argument 1.
<i>First()</i> , <i>Last()</i>	Get the <i>first</i> and <i>last</i> element of a list or character string.
<i>Strip(list, n)</i>	Strip the nth element from the list. (n=0,1,2...)
<i>Len()</i>	Get the <i>length</i> of a list or character string.
<i>Sort()</i>	Sort all elements in a list from small to large.
<i>Reverse()</i>	Reverse the order of a list.
<i>TopLo()</i> , <i>TopHi()</i>	Sort all elements in a list from small to large; large to small.
<i>Dim()</i>	Return an integer list of dimension lengths. Zero value is end of array.
<i>Det()</i>	Get the determinant of a matrix

1.2 Concatenation of objects

The operators + and * may be applied to string objects. Here is an example of the + operator.

```
s1 = "this" ; s2 = " and that"
s3 = s1 + s2
```

The object s3 now holds "this and that".

Strings made up of multiple instances of the same character sequence can be generated by the * operator, as below.

```
s4 = "-"*10
```

The object s4 now holds a string "-----". The * operator can be applied to named objects as well.

```
s4 = "-EOF-" ; s5 = s4*3
```

The object s4 now holds a string "-EOF--EOF--EOF-".

2. LIST MEMBERSHIP

It is possible to test objects containing lists of strings for the "membership" of specific strings. These tests are equivalent to looping through the lists and applying the standard string equivalence operators "==" and "!=", as illustrated in the following example statements.

```
cnt = List(["data_", "global_", "save_", "stop_", "loop_"])
Do i=0,4 { If("stop_" == cnt[i]) Break ;}
```

The last statement is problematical because the length of the list of items being tested needs to be known. It may be replaced simply by:

```
If ("stop_" in cnt) { ... }
```

This works only if elements of the container are of the same type. The negation test for membership of a list also applies. E.g.

```
If ("cell_" not in cnt) { ... }
```

3. LIST NOTATION

The following notation is available for the formation of lists from existing named lists.

<i>new</i> = <i>list</i> [:]	New copy of entire list.
<i>new</i> = <i>list</i> [<i>n:m:i</i>]	New list with elements from indices <i>n</i> to <i>m</i> in steps of <i>i</i> .
<i>new</i> = <i>list</i> [<i>n:m</i>]	New list of elements from indices <i>n</i> to <i>m</i> in steps of 1.
<i>val</i> = <i>list</i> [1]	<i>val</i> becomes the value of the second element of <i>list</i> .
<i>new</i> = <i>list1</i> + <i>list2</i>	New list of <i>list1</i> concatenated with <i>list2</i> .
<i>new</i> = [<i>list1</i> , <i>list2</i>]	New list of <i>list1</i> concatenated with <i>list2</i> .
<i>val1</i> += <i>val2</i>	Increment <i>val1</i> with <i>val2</i> .
<i>list1</i> += <i>val</i>	Increment all elements in <i>list1</i> with <i>val</i> .
<i>list1</i> += <i>list2</i>	Increment matching elements in <i>list1</i> with values in <i>list2</i> .
<i>list1</i> ++= <i>val</i>	Append <i>val</i> to <i>list1</i> .
<i>list1</i> ++= <i>list2</i>	Append <i>list2</i> to <i>list1</i> .
<i>val1</i> -= <i>val2</i>	Decrement <i>val1</i> with <i>val2</i> .
<i>list1</i> -= <i>val</i>	Decrement all elements in <i>list1</i> with <i>val</i> .
<i>list1</i> -= <i>list2</i>	Decrement matching elements in <i>list1</i> with values in <i>list2</i> .
<i>list</i> [<i>i:j</i>] = <i>list2</i>	Cut and paste ALL of <i>list2</i> into the elements <i>i</i> to <i>j</i> -1.

4. ARRAY NOTATION

The following notation applies strictly to *Array* objects.

$var = mat[n,m]$	Variable contains the value of the matrix element (n,m)
$vec = vec1 + val$	Scalar addition. $[9,10,11] = Vector([4,5,6]) + 5$
$vec = Function(vec1)$	Vector function. $[1,2,0] = Mod([4,5,6], 3)$ for $(Mod, Int,)$
$vec = vec1 + vec2$	Vector addition. $[12,14,16] = Vector([4,5,6]) + Vector([8,9,10])$
$var = vec1 * vec2$	Scalar (dot) product. $8*4+9*5+10*6 = Vector([4,5,6]) * Vector([8,9,10])$
$vec = vec1 ^ vec2$	Vector (cross) product. $(-4,8,-4) = Vector([4,5,6]) ^ Vector([8,9,10])$
$vec = mat * vec1$	Post-matrix vector multiply. E.g. $[32,77,112] = Matrix([[1,2,3],[4,5,6],[7,8,9]]) * Vector([4,5,6])$
$vec = vec1 * mat$	Pre-matrix vector multiply. E.g. $[66,81,96] = Vector([4,5,6]) * Matrix([[1,2,3],[4,5,6],[7,8,9]])$
$mat = mat1 * mat2$	Matrix multiply. Matrices must have concordant shapes.